

Plotting a Projectile

David Grosjean

In this issue we will compare programming the VZ200, the color and sound computer by Video Techonology, and the TS1000. The project we will undertake is the plotting of a projectile.

Starting with a Clear Screen

Let's start with a simple clear screen and plot statement.

TS1000:

```
10 CLS
200 PLOT X,Y
```

VZ200:

```
5 CLS
40 MODE(1):COLOR 4
200 SET(X,Y)
```

If you look at the VZ200 program, you will notice that the computer has to be put into a special graphics mode with line 40. This means that you cannot have the medium resolution graphics and text on the screen at the same time. This will become a problem when we try to turn this into a game.

The Projectile Equations

The equations for the horizontal and vertical position of a projectile are:

$$X = V \cdot \cos(A) \cdot T$$

$$Y = V \cdot \sin(A) \cdot T - 1/2 \cdot G \cdot (T \cdot T)$$

V is the velocity; T is the time; G is the effect of gravity. These equations can be worked into the program like this:

TS1000:

```
20 LET V=1000
30 LET D=57.3
40 LET A=45
50 LET C=V*SIN(A/D)
60 LET C1=V*COS(A/D)
80 FOR T=0 TO 44 STEP .5
90 LET X=C1*T
100 LET Y=C*T-16*T*T
150 LET X=X/500
150 LET Y=Y/500
220 NEXT T
```

VZ200:

```
10 A=45
20 V=1000:G=32
30 D=57.3
50 C=V*SIN(A/D)
60 C1=V*COS(A/D)
```

```
80 FOR T=0 TO 45 STEP .5
90 X=C1*T
100 Y=C*T-16*T*T
180 X=X/250
190 Y=Y/250
220 NEXT T
```

As you will notice, the range on the VZ200 increased due to the higher resolution of the graphics, but we did not change the velocity of the projectile. Instead, we changed the number which we divide X and Y by to fit the projectile on the different screen size.

In these programs, D is a factor that converts degrees to radians which are what the computer wants. C and C1 are constants for each firing angle. When you RUN this program on the VZ200, you will notice that the plot is upside down. This is because the vertical distances are measured from top to bottom instead of bottom to top as on the TS1000. Change line 190 in the VZ200 program to

```
190 Y=61-Y/250
```

Setting the Gun Angle

Now we can modify the programs to accept a gun angle from 1 to 90 degrees.

TS1000:

```
40 PRINT "ANGLE OF GUN?"
45 INPUT A
70 LET T1=2*C/32
80 FOR T=0 TO T1 STEP .5
230 GOTO 50
```

VZ200:

```
10 INPUT "ANGLE OF GUN";A
70 T1=2*C/G
80 FOR T=0 TO T1 STEP .5
230 GOTO 50
```

Making a Game

Now that we have a working, however simple, projectile program, let's try to make a game out of it. The following games are our projectile programs tightened up a bit and with the provisions for a target.

Setting up the Target

On the VZ200 the range is 127,000 yards, and on the TS1000 32,000 yards (1000 yards for every horizontal position on the screen). This will throw the equation off a little since the gun cannot shoot the

November/December 1983 • SYNC

projectile 127,000 yards. (If this bothers you, think of the yards on the VZ200 as 11-inch feet.)

Although there are 64 pixel positions on the TS1000, the target is a T which takes up two pixels. You can hit the left or the right of the T so the number of effective horizontal positions is reduced to half. Notice that, since the VZ200 cannot have text and graphics on the screen at once, line 100 forms a special target, while on the TS1000, a simple PRINT AT command in line 60 does the same thing.

TS1000:

```
20 LET V=1000
40 LET K=INT (20000*RND)+12000
50 CLS
52 PRINT "RANGE = 32000 YDS"
60 PRINT AT 21,INT (K/1000);"T"
70 PRINT AT 1,0;"ANGLE OF GUN?"
80 INPUT A
90 IF A<1 OR A>90 THEN GOTO 90
120 LET C=V*SIN (A/57.3)
130 LET C1=V*COS (A/57.3)
140 LET T1=2*C/32
150 FOR T=0 TO T1 STEP .5
160 LET X=C1+T/500
170 LET Y=T*(C-16*T)/500
180 PLOT X,Y
190 NEXT T
```

VZ200:

```
20 V=1000
40 K=INT (97000*RND(0))+30000
50 PRINT "RANGE = 127000 YDS"
60 PRINT "TARGET AT";K;"YDS"
70 INPUT "ANGLE OF GUN";A
80 IF A<1 OR A>89 THEN 70
90 MODE(1):COLOR4
100 FOR L=1 TO 4:FOR L1=1 TO 4:SET
  (INT(K/1000-A)+L1,59+L):NEXT: NEXT
130 C=V*SIN(A/57.3)
140 C1=V*COS(A/57.3)
150 T1=2*C/32
160 FOR T=0 TO T1 STEP .5
170 X=C1+T/250
180 Y=61-(T*(C-16*T)/250)
190 SET(X,Y)
210 GOTO 210
```

Detecting a Hit

We now have a target, but it is of no use unless the computer can detect its destruction. The following lines detect a hit. Notice how the techniques of detecting a hit target differ. The VZ200 must compare each position of the target, which is four positions wide, with the last position of the projectile; the TS1000 does the same thing but uses the PRINT AT position used by the target to compare to the last position of the projectile. This is, of course, simpler. Line 300 in the VZ200 version is a special "explosion" accompanied with some sounds. You can experiment at this point to find a better explosion.

TS1000:

```
300 IF INT (X/2)=INT (K/1000) T
HEN GOTO 300
250 GOTO 50
300 PRINT AT 21,INT (K/1000)-2;
310 PAUSE 250
340 GOTO 30
```

VZ200:

```
220 FOR L=1 TO 4:IF INT(K/1000)-
L=INT(X) THEN 300
```

```
225 NEXT L
250 GOTO 50
300 FOR L=1 TO 30:SET(40+87*RND(0
),40+22*RND(0)):SOUND31,1:NEXT L
310 PRINT "HIT! HIT! HIT!"
340 GOTO 30
```

Making the Next Shot

Now we can add the response the computer will make to a missed target. The following lines tell how far away your shot was from the target and lets you try again. Line 210 in the VZ200 version is a delay loop so you have time to see the last position of the projectile.

TS1000:

```
210 LET E=INT (K-(32000*SIN (.0
35+R)))
220 IF E<100 THEN PRINT AT 0,0;
"OVER BY";ABS E;" YDS"
230 IF E>100 THEN PRINT AT 0,0;
"UNDER BY";ABS E;" YDS"
240 PAUSE 250
```

VZ200:

```
210 FOR L=1 TO 3000:NEXT L
230 IF INT(K/1000)>X THEN PRINT
"UNDER BY";K-X*1000;"YDS"
240 IF INT(K/1000)<X THEN PRINT
"OVER BY";X*1000-K;"YDS"
```

Providing Your Shots

The computer can now detect hits and misses. This is where the game part comes in. The following lines provide you with 5 individual targets with a maximum of 5 attempts to hit each target. If you fail to hit a target in 5 shots, you lose. S is the number of shots you have taken per target; S1 is your total number of shots; and Z is the total number of targets.

TS1000:

```
5 LET Z=0
10 LET S1=0
30 LET S=0
55 IF S=5 THEN GOTO 260
100 LET S1=S1+1
110 LET S=S+1
260 PRINT AT 0,0;"ENEMY GOT YOU
FIRST"
270 GOTO 370
320 LET Z=Z+1
330 IF Z=5 THEN GOTO 350
```

VZ200:

```
10 S1=0:Z=0
30 S=0
55 IF S=5 THEN 260
110 S=S+1
120 S1=S1+1
260 PRINT "THE ENEMY GOT YOU FIR
ST!"
270 GOTO 370
320 Z=Z+1
330 IF Z=5 THEN 350
```

Evaluation and Restart

Finally, we need an evaluation and a mechanism to restart the game. The following lines do this.

TS1000:

```
350 PRINT AT 0,0;S1;" ROUNDS US
ED"
355 IF S1<10 THEN PRINT "GREAT
JOB"
360 IF S1>15 THEN PRINT "YOU CA
N DO BETTER"
370 PRINT "PLAY AGAIN?"
380 INPUT Z$
390 IF Z$="Y" THEN RUN
```

VZ200:

```
350 PRINT S1;"ROUNDS USED"
355 IF S1<10 THEN PRINT "GREAT J
OB!"
360 IF S1>15 THEN PRINT "YOU COU
LD HAVE DONE BETTER"
370 INPUT "PLAY AGAIN";Z$
380 IF Z$="Y" THEN RUN
```

Improving on the Game

Of course, these artillery-type games are very simple. They provide a basic game which you can elaborate on or experiment with to develop different possibilities. You might want to improve on the graphics or sound on the VZ200 or perhaps make a really BIG explosion. Although the TS1000 has no color or sound, the program can still be greatly improved. You could add hi-res graphics through either a hardware add-on or a software program. You might want to add a sound unit which will give the sound effects or add a routine to provide some sound (e.g., AUDISY).

